Problem:
Score : 300 points

Problem Statement

There are N integers, A_1, A_2, ..., A_N, written on the blackboard.

You will choose one of them and replace it with an integer of your choice between 1 and 10^9 (inclusive), possibly the same as the integer originally written.

Find the maximum possible greatest common divisor of the N integers on the blackboard after your move.

Constraints

All values in input are integers.

2 \leq N \leq 10^5

1 \leq A_i \leq 10^9

Output

Input is given from Standard Input in the following format:

N
A_1 A_2 ... A_N

Output

Print the maximum possible greatest common divisor of the N integers on the blackboard after your move.

Sample Input 1

3
7 6 8

Sample Output 1

2

If we replace 7 with 4, the greatest common divisor of the three integers on the blackboard will be 2, which is the maximum possible value.

Sample Input 2

3
12 15 18

Sample Output 2

6

Sample Input 3

2
1000000000 1000000000

Sample Output 3

1000000000

We can replace an integer with itself.

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program main
  implicit none
  integer(8) :: i, j, n, result = 0
  integer(8), allocatable :: a(:), l(:), r(:)

  read *, n
  allocate (a(n), l(0:n), r(n+1))
  read *, a

  l(0) = 0
  l(1) = a(1)
  r(n+1) = 0
  r(n) = a(n)
  do i = 2, n
    l(i) = euclid(l(i-1), a(i))
    r(n+1-i) = euclid(r(n+2-i), a(n+1-i))
  end do

  do i = 1, n
    result  = max(euclid(l(i-1), r(i+1)), result)
  end do

  print '(i0)', result

contains
  function euclid(a, b) result (res)
    integer(8) :: a, b, u, v, res
    u = a; v = b
    do while(v /= 0)
      res = mod(u, v)
      u = v
      v = res
    end do
    res = u
  end function
end program